



钣金工下料电算法

金枫天 编著

浙江科学技术出版社

板金工下料电算法

金枫天 编著

浙江科学技术出版社

责任编辑：骆 健
封面设计：潘孝忠



板金工下料电算法

金枫天 编著

*

浙江科学技术出版社出版

浙江新华印刷厂印刷

浙江省新华书店发行

开本：787×1092 1/16 印张：9.75 字数：238,000

1983年6月第一版

1983年6月第一次印刷

印数：1-7,000

统一书号：15221·39

定 价：1.04 元

前 言

电子计算机是本世纪的重大科研成果。它的出现,对科学技术的各个领域,特别是对计算数学产生了深远的影响,并且在各行各业中得到了越来越广泛的应用。

电子计算机近年来在我国也已逐渐推广应用。为了使电子计算机在社会主义经济建设中发挥更大的作用,编者在过去编写的《常用板金工计算下料手册》的基础上,结合实际应用,又编写了《板金工下料电算法》一书。

本书各章节程序均用BASIC的十七种基本语言编写,并尽可能做到程序的多用性和在各类电子计算机上的通用性。对一些由于计算机内在功能的原因而容易发生溢出的计算式也作了改进,即按泰勒级数展开来计算。这样既保证了运算精度,又确保了计算机的正常运行。为慎重起见,本书各章节程序均在CROMEMCO微型计算机上通过,结果正确无误。本书各章节的解析计算式都是经过严密的数学推导的,有些计算式还修正了以往的一些近似计算式。

本书不但能迅速无误地计算出各种板金制件的展开图尺寸,而且还可作为一些初学BASIC语言的读者的一本实际应用参考书。通过该书的学习,可以初步掌握用BASIC语言编制程序的方法。随着程序外存技术的发展,在有电子计算机的单位,可事先将程序外存在磁盘、纸带或磁带中。这样,在具体应用时就不必再打印程序,更显示了本书在实际应用中的优越性。

本书初稿写成后,浙江大学数学系余奕岳老师作了详细的审阅及修改,并提出了许多宝贵意见,特在此表示衷心的感谢。由于编者水平有限,编制程序经验不足,书中如有贻误之处,深望广大读者批评指正。

1982年12月

目 录

第一章 总论	(1)
第二章 等径圆柱形管件	(4)
第一节 不同节数、任意角度圆管弯头	(4)
第二节 斜交三通管 (包括垂直〈正交〉三通管〈T形〉)	(5)
第三节 任意角度三通管 (Y形)	(7)
第四节 长短脚任意角度裤形三通管 (包括等长脚任意角度裤形三通管)	(9)
第五节 带补料的垂直三通管 (T形) (包括带补料的垂直四通管)	(11)
第六节 带补料的120°三通管 (Y形)	(13)
第七节 带补料的任意角度三通管 (Y形)	(15)
第八节 任意角度四通管 (包括垂直〈正交〉四通管)	(17)
第九节 任意角度蛇形弯管	(20)
第十节 双扭向90°和任意角度蛇形弯管 (包括双扭向双90°蛇形弯管)	(22)
第十一节 四节双向90°弯头三通管	(24)
第十二节 带斜插圆管的四节90°圆管弯头	(26)
第三章 异径圆柱形管件及异口形管件	(29)
第一节 任意角度斜交不等径三通管 (包括垂直相交不等径三通管)	(29)
第二节 一侧垂直相接 (移心的) 不等径三通管	(33)
第三节 矩形管与圆管垂直相交三通管	(36)
第四节 矩形管与圆管任意角度斜接三通管	(37)
第五节 扩径三通管	(40)
第六节 圆管与矩形管相接的连接管 (包括圆管与方管相接的连接管)	(42)
第七节 圆管与矩形管移心相接的连接管	(44)
(1. 在矩形长的垂直平分线上移心, $L_1 > L_2$)	
(2. 在矩形宽的垂直平分线上移心, $L_1 < L_2$)	
第八节 圆管与方管移心相接的连接管	(47)
第九节 圆管与矩形管向任意方向移心相接的连接管	(49)
第十节 上方下圆V形三通管	(52)
第十一节 椭圆	(54)
第十二节 椭圆大小头	(55)
第十三节 椭圆管与圆管相接的连接管	(57)

第十四节	椭圆管与圆管移心相接的连接管	(59)
	(1. 在椭圆的长轴上移心, $A > B$, 2 A 为长轴, 2 B 为短轴)	
	(2. 在椭圆的短轴上移心, $A < B$, 2 B 为长轴, 2 A 为短轴)	
第四章	圆锥、圆台、移心管件	(63)
第一节	正圆锥	(63)
第二节	正圆台 (亦称同心大小头)	(64)
第三节	斜切圆锥台 (底面圆为大圆)	(66)
第四节	斜切圆锥台 (底面圆为小圆)	(68)
第五节	任意角度两节渐缩弯头	(70)
第六节	任意角度四节渐缩弯头 (包括 90° 四节渐缩弯头)	(74)
第七节	移心大小头	(79)
第八节	圆锥台——圆管两节任意角度弯头 (圆锥台底面圆为大圆)	(81)
第九节	圆锥台——圆管两节任意角度弯头 (圆锥台底面圆为小圆)	(85)
第十节	圆管——移心大小头——圆管裤形三通管	(89)
第十一节	异径的 90° 连接管	(92)
第十二节	放射形渐缩四通管	(95)
第十三节	Y 形渐缩三通管	(97)
第五章	棱锥、棱台管件	(100)
第一节	正 N 棱锥	(100)
第二节	正 N 棱台	(102)
第三节	矩形棱台	(104)
第四节	移心矩形棱台	(105)
第五节	任意倾角斜截上口棱台 (斜截后上、下口为正方形)	(107)
第六节	上下口扭转 90° 的矩形连接管	(108)
第七节	任意倾角斜截正四棱台	(109)
第八节	上下口扭转 45° 的正方形渐缩管	(110)
第九节	任意角度两节渐缩方口弯头	(111)
第十节	三节 90° 渐缩方口弯头	(114)
第十一节	大小方口渐缩三通管	(118)
第十二节	圆管直交正四棱台	(120)
第十三节	大小方口移心渐缩裤形三通管	(121)
第十四节	凸五角星	(124)
第六章	方口曲面管、球体、长圆管	(125)
第一节	矩形管弧面任意角度弯头	(125)
第二节	方口弧面渐缩成矩形管的 90° 弯头	(126)
第三节	方口弧面罩	(128)
第四节	方管直交方口弧面罩	(129)
第五节	矩形口弧面罩	(131)
第六节	方口蛇形连接管	(133)

第七节	螺旋输送机的回转叶片(又名绞龙)	(134)
第八节	矩形管渐缩成方管的90°弯头(矩形口宽等于方口边长)	(135)
第九节	绕圆柱螺旋上升360°的矩形管	(137)
第十节	球表面体(一)	(139)
第十一节	球表面体(二)	(140)
第十二节	渐缩长圆管	(142)
第十三节	不同节数、任意角度长圆管弯头(立弯形)	(143)
第十四节	不同节数、任意角度长圆管弯头(平弯形)	(145)
第十五节	垂直(正交)长圆三通管(立交形)	(146)
第十六节	垂直(正交)长圆三通管(平交形)	(148)

第一章 总 论

电算法,顾名思义即为用电子计算机计算的方法。本书就是应用电子计算机来计算各种类型板金制品的展开图的尺寸,使得各种复杂的板金制品的展开图计算只要花几秒钟即可完成。

目前,由于国内大多数中小型电子计算机都配有 BASIC 语言(BASIC 是 BEGINNER'S ALL-PURPOSE SYMBOLIC INSTRUCTION CODE 一词的缩写,即初学者通用符号指令代码),而 BASIC 语言又比较简单易学。BASIC 的基本语句只有17种,它所使用的命令和语句中使用的词、运算符与英语中使用的词以及数学中的符号差不多,比较直观,易于理解。例如:“PRINT”的意思是“打印”,在 BASIC 语言中它也代表“打印”,它使计算机打印出所需要的结果。“END”的意思是“结束”,在 BASIC 语言中,它就表示本程序结束,计算机接到此命令就停止工作。同时,BASIC 语言能比较直接地改写为 FORTRAN 语言,而 FORTRAN 语言目前在国内使用比较广泛。所以本书采用 BASIC 的基本语言编写(还有扩展 BASIC 和多用户分时 BASIC,由于本书计算不需要,未采用)。

一个工程实际问题,用电子计算机解算的全过程,包括如下几个步骤:

一、对实际问题用一系列数学式子来表示,即建立一个数学模型。

二、对数学模型选择一个比较合理的计算方法,即采用何种算法语言。

三、根据所采用的算法语言编写程序,即确定解题顺序,并按电子计算机所能接受的方式描述出来。这种描述叫做“程序”,程序由各种语句组合而成。

四、上机计算。在计算时,可通过电传打字机键盘直接输入进行计算,也可使用磁盘、磁带*来外存程序。这样每次计算时,只要根据编码的顺序调用某程序,不必再将该程序在电传打字机键盘上打印一遍。因而可快速无误地进行各种计算(*用磁带记录数据是由杭州大学数学系试制成功的,它具有传输速度快,记录密度高,可靠性强,接口简单,使用方便等优点)。

本书每节从使用方便和直观出发,按以下格式编写:

一、示意图:包括视图和展开图,起各个字母符号所表示的尺寸线位置的指示作用。

二、程序及例题:包括用 BASIC 的基本语言编写的各种程序,并按各假定尺寸计算运行的例题。每节都附有例题的目的,是为了使读者在应用电传打字机键盘直接输入程序时,可检验打印是否有误。因为本书每节的程序都由 CROMEMCO 微型机运算检验过,当使用者每次打出 RUN 后,计算机打印出?时,第一次可按书上所给的尺寸输入,如果计算机字符显示器上显示出的值与书上的值一样,那么说明键盘输入的程序无误,可放心进行算题运算,并在电传打字机上打印出结果(程序和检验都不必在电传打字机上打印出来,这样可节约计算机用纸)。

三、使用说明及字母符号的意义:指出编制程序和调用过程时应注意的事项,并给出各已知数字母符号的实际含义,使读者不致与计算结果相混淆。

四、基本计算公式：在直角坐标下给出各个基本参数的计算公式或解析式，这与具体应用无直接关系，主要供读者碰到其他板金制品或组合形式的板金制品的展开图的计算时，可以参考这些基本计算公式来编制新的程序。为了避免喧宾夺主，这些基本计算公式的推导过程，本书不作介绍。读者如有兴趣，可参看浙江科学技术出版社1981年8月出版的《常用板金工计算下料手册》一书，该书对各节的基本计算公式有较详细的推导过程。

本书第四章第八节、第九节的解析式是经过严密的数学推导的，修正了以往一些书上的近似计算公式，使这两节的大型板金制品的精度大大的提高了。

本书根据各类板金制品的外形特点分为六章，并且各章节程序均在CROMEMCO微型电子计算机上实算通过。例题中的各计算值也是按打印结果抄录的。但是每一种计算机具体所用的BASIC语言规定还有某些细小的区别，如CROMEMCO微型机的赋值语句前可省略LET，打印语句不用PRINT表示而用符号“@”表示。本书为清楚起见，在赋值语句前仍加上LET，打印语句仍写为PRINT。因此在使用前，必须查阅一下你所使用的计算机系统的BASIC说明书和该计算机的操作说明。

本书各章节大多数在INPUT语句中需输入等分数(G)。因为斜切圆管展开后，斜切口是一个周期的正弦曲线。斜切圆锥展开后，斜切口也是一条曲线。如何正确作出这条曲线是作展开图的关键。先将圆管（或圆锥等）展开后的周长分成若干等分，用计算机算出各等分上对应点的数值，并在展开图上标出。然后依次连接各点，就得到所要作的近似曲线。等分越多，得到的近似曲线的精度越高，但相应要计算的数值也越多，作图也越复杂。一般可以根据圆管半径的大小来决定所取的等分数（对于圆锥台、渐缩弯管等可按大头半径来定，对于不等径管件按大圆管半径定）。具体见下表。

等 分 数 表

圆管半径R(mm)	等 分 数 (G)	圆管半径R(mm)	等 分 数 (G)
50以内	8	400~650	32
50~150	12	650~1000	48
150~250	16	1000~2000	64
250~400	24	2000以上	96

一般在INPUT语句中输入的圆管半径(R值)是指中径。但如果已知圆管外半径(R')，且钢板厚度为t时，则INPUT语句中输入的圆管半径(R值)应等于 $\left(R' - \frac{t}{2}\right)$ 。如果已知圆管内半径(R'')，则INPUT语句中输入的圆管半径(R值)应等于 $\left(R'' + \frac{t}{2}\right)$ 。

当钢板厚度 $t > 3\text{mm}$ 时（以两节圆管弯头为例），从图1—1可以看出，两圆管弯头的最高接点为O，最低接点为A，K为反曲点。从A点到K点是外皮相接，从K点到O点是里皮相接，因此必须进行板厚处理。如不进行板厚处理，两节圆管弯头的夹角定会发生变化。而当节数愈多时，夹角的误差也愈大，会严重地影响板金制品的精确度。

为了使两节圆管弯头的夹角不产生误差，必须保证两节圆管弯头的接触点都是外皮相

接, 因为从A点到反曲点K内的曲线都是凹的, 而且都是外皮相接。因此, 从A点到反曲点K内的曲线(也就是所有凹的曲线), 不需要进行板厚处理。而从反曲点K到O点内的曲线都是凸的, 而且都是里皮相接, 因此, 从反曲点K到O点内的曲线(也就是所有凸的曲线), 都必须进行板厚处理。




图 1—1




图 1—2

板厚处理法: 将图 1—1 其中一节展开, 如图 1—2 所示。因为钢板厚 $t > 3\text{mm}$, 所以只能用气割进行下料。在曲线凹的地方(从A点到反曲点K内), 割枪头垂直钢板进行切割。在曲线凸的地方(从反曲点K到O点内), 割枪头沿曲线向水平线倾斜切割。倾斜角度应不小于 α 度。另一节圆管也按同样方法进行板厚处理。本书以下各章节均按以下原则和方法进行板厚处理: 所有凹的曲线不需进行板厚处理, 所有凸的曲线必须进行板厚处理。

在有微型电子计算机的企业单位或计算站, 最好能事先将本书各章节的程序外存到磁盘、纸带或磁带里, 这样每当实际工程需要计算时, 可迅速、正确地给出计算结果。

第二章 等径圆柱形管件

第一节 不同节数、任意角度圆管弯头

一、示意图：





图 2—1




图 2—2

二、程序及例题：

```

10 PRINT "R, R1, B, N, G=";
20 INPUT R, R1, B, N, G
30 LET A=6.28318×R
40 LET M=A/G
50 LET B=B×3.14159/180
60 LET B=B/(2×(N-1))
70 LET C=(R1-R)×TAN(B)
80 PRINT "A="; A, "M="; M, "C="; C
90 FOR I=1 TO G/2
100 LET X=I×6.28318/G
110 LET E=R×(1-COS(X))×TAN(B)
120 PRINT "E(";I;")="; E
130 NEXT I
140 END
RUN

```

R, R1, B, N, G=? 86, 150, 90, 4, 12

A=540.35348 M=45.029456666667 C=17.148733147006

$E(1)=3.087253271671$
 $E(2)=11.521787431147$
 $E(3)=23.043579592181$
 $E(4)=34.565379945348$
 $E(5)=42.99993648669$
 $E(6)=46.087220332498$

*** 140 END ***

三、使用说明及字母符号的意义:

本程序用 INPUT 语句输入已知数:

R: 表示圆管半径。

R1: 表示弯头的中心半径。

B: 表示弯头的角度。

N: 表示节数。

G: 表示等分数 (具体数值见第一章等分数表。以下各章节的 G 值均参照等分数表确定, 不再复述)。

算出的各值均为 I、N 展开图的值。I、... II (N-1)。

各展开图为 I 展开图的两倍。具体下料可参照排料图 (图 2-2)。

注意: 本程序输入的 B 值为角度。

四、基本计算公式:

$$A = 2 \cdot \pi \cdot R \quad M = \frac{2 \cdot \pi \cdot R}{G} \quad B = \frac{B}{2(N-1)} \quad B = \frac{B \cdot \pi}{180} \quad \text{化成弧度}$$

$$C = (R1 - R) \cdot \tan B$$

$$X = \frac{2 \cdot \pi \cdot I}{G}, \dots, \pi \quad I = 1, 2, \dots, \frac{G}{2} \quad E(I) = R \cdot (1 - \cos X) \cdot \tan B$$

第二节 斜交三通管

(包括垂直<正交>三通管<T形>)

一、示意图:




图 2-3




图 2—4

二、程序及例题，

```

10 PRINT "R, B, L1, L2, L3, G=";
20 INPUT R, B, L1, L2, L3, G
30 LET A=6.28318×R
40 LET M=A/G
50 PRINT "A="; A, "M="; M
60 LET B=B×3.14159/180
70 LET D=B/2
80 FOR I=1 TO G/4
90 LET X=I×6.28318/G
100 LET E=R×(1-COS(X))×COS(D)/SIN(D)
110 LET S=R×(1-COS(X))×TAN(D)
120 PRINT "E("; I; ")="; E, "S("; I; ")="; S
130 NEXT I
140 END

```

RUN

R, B, L1, L2, L3, G=? 230, 62, 600, 200, 1200, 16

A=1445.1314 M=90.3207125

E(1)=29.137699986345 S(1)=10.519663547549

E(2)=112.11485665469 S(2)=40.477133447104

E(3)=236.29897115595 S(3)=85.311664076349

E(4)=382.78416931772 S(4)=138.19761595587

*** 140 END ***

三、使用说明及字母符号的意义：

本程序用 INPUT 语句输入已知数：

R：表示圆管半径。

B：表示斜交角度。（ $B \leq 90^\circ$ ）

L1、L2、L3, 均表示示意图上的实长。

G: 表示等分数。

算出的各值适合于 I、II 展开图, 具体可按示意图(图 2—3)划线下料。

注意: ①本程序输入的 B 值为角度。

②当 $B = 90^\circ$ 时, 斜交三通管即变成垂直(正交)三通管(T形)。因此, 可用本程序来计算垂直(正交)三通管(T形)展开图的尺寸。算出各值后, 可按示意图(图 2—4)划线下料。

四、基本计算公式:

$$A = 2 \cdot \pi \cdot R \quad M = \frac{2 \cdot \pi \cdot R}{G} \quad B = \frac{B \cdot \pi}{180} \quad \text{化成弧度} \quad D = \frac{B}{2}$$

$$X = \frac{2 \cdot \pi \cdot I}{G}, \dots, \frac{\pi}{2}$$

$$E(I) = \frac{R \cdot (1 - \cos X)}{\operatorname{tg} D} \quad S(I) = R \cdot (1 - \cos X) \cdot \operatorname{tg} D$$

第三节 任意角度三通管(Y形)

一、示意图:




图 2—5

二、程序及例题:

```

10 PRINT "R, B, L1, L2, G=";
20 INPUT R, B, L1, L2, G
30 LET A = 6.28318 * R
40 LET M = A / G
50 PRINT "A="; A, "M="; M
60 LET B = B * 3.14159 / 180
70 FOR I = 1 TO G / 4
80 LET X = I * 6.28318 / G
90 LET E = R * (1 - COS(X)) * COS(B / 2) / SIN(B / 2)

```

```

100 LET      S=R*(1-COS(X))*TAN(B/4)
110 PRINT    "E("; I; ")="; E  "S("; I; ")="; S
120 NEXT     I
130 END

```

RUN

R, B, L1, L2, G=? 280, 100, 800, 800, 16

A=1759.2904 M=109.95565

E(1)=17.884340683329 S(1)=9.9387298041794

E(2)=68.814638527224 S(2)=38.241840222374

E(3)=145.03723029798 S(3)=80.60044644653

E(4)=234.94793670987 S(4)=130.56584541501

*** 130 END ***

三、使用说明及字母符号的意义:

本程序用 INPUT 语句输入已知数:

R: 表示圆管半径。

B: 表示两节相同圆管 I 的夹角。

L1、L2: 均表示示意图上的实长。

G: 表示等分数。

算出各值后可按示意图 (图 2—5) 划线下料。

注意: 本程序输入的 B 值为角度。

四、基本计算公式:

$$A = 2 \cdot \pi \cdot R \quad M = \frac{2 \cdot \pi \cdot R}{G} \quad B = \frac{B \cdot \pi}{180} \quad \text{化成弧度}$$

$$X = \frac{2 \cdot \pi \cdot I}{G}, \dots, \frac{\pi}{2} \quad I = 1, 2, \dots, \frac{G}{4}$$

$$E(I) = \frac{R \cdot (1 - \cos X)}{\operatorname{tg} \frac{B}{2}}$$

$$S(I) = R \cdot (1 - \cos X) \cdot \operatorname{tg} \frac{B}{4}$$

第四节 长短脚任意角度裤形三通管

(包括等长脚任意角度裤形三通管)

一、示意图:




图 2-6

二、程序及例题:

```

10 PRINT "R, B, L1, L2, L4, L5, G=";
20 INPUT R, B, L1, L2, L4, L5, G
30 LET A=6.28318*R
40 LET M=A/G
50 LET B=B*3.14159/180
60 LET L3=L2+(L4-L5)*COS(B/2)/SIN(B/2)
70 LET L6=L4/SIN(B/2)+R*TAN(B/4)
80 LET L7=L5/SIN(B/2)+R*TAN(B/4)
90 PRINT "A="; A, "M="; M,
100 PRINT "L3="; L3, "L6="; L6, "L7="; L7
110 FOR I=1 TO G/2
120 LET X=I*6.28318/G
130 IF I>G/4 THEN 170

```

```

140 LET      E=R*(1-COS(X))*COS(B/2)/SIN(B/2)
150 LET      S=R*(1-COS(X))*TAN(B/4)
160 PRINT    "E("; I; ")="; E, "S("; I; ")="; S
170 LET      H=R*(1-COS(X))*TAN(B/4)
180 PRINT    "H("; I; ")="; H
190 NEXT     I
200 END

```

RUN

R, B, L1, L2, L4, L5, G=? 300, 110, 900, 600, 1050, 750, 16

A=1884.954 M=117.809625 L3=810.06262397015

L6=1437.9840065226 L7=1071.7514219687

E(1)=15.990038473997 S(1)=11.88771058375

H(1)=11.88771058375

E(2)=61.525819548394 S(2)=45.741049179386

H(2)=45.741049179386

E(3)=129.67494489682 S(3)=96.406160460719

H(3)=96.406160460719

E(4)=210.06234526046 S(4)=156.16975337871

H(4)=156.16975337871

H(5)=215.93337784175

H(6)=266.59857895532

H(7)=300.4520519955

H(8)=312.33992116739

*** 200 END ***

三、使用说明及字母符号的意义:

本程序用 INPUT 语句输入已知数:

R: 表示圆管半径。

B: 表示圆管 I 与圆管 (II) 的夹角。

L1、L2、L4、L5: 均表示示意图上的实长。

G: 表示等分数。

算出各值后按示意图 (图 2-6) 划线下料。

注意: ①本程序输入的 B 值为角度。

②当 L4=L5 时, 长短脚任意角度裤形三通管即为等长脚任意角度裤形三通管。

因此, 可用本程序来计算等长脚任意角度裤形三通管展开图的尺寸, 算出各值后, 可按示意图 (图 2-6) 划线下料。

四、基本计算公式:

$$A = 2 \cdot \pi \cdot R \quad M = \frac{2 \cdot \pi \cdot R}{G} \quad B = \frac{B \cdot \pi}{180} \quad \text{化成弧度}$$